%%%-------------------------------------------------------------------
%%% File    : p18.erl
%%% Author  : Plamen Dragozov <plamen at dragozov.com>
%%% Description : 
%%% By starting at the top of the triangle below and moving 
%%% to adjacent numbers on the row below, the maximum total from 
%%% top to bottom is 23.
%%%    3
%%%   7 5
%%%  2 4 6
%%% 8 5 9 3
%%%
%%% That is, 3 + 7 + 4 + 9 = 23.
%%%
%%% Find the maximum total from top to bottom of the triangle below:
%%%                 75
%%%               95 64
%%%              17 47 82
%%%             18 35 87 10
%%%            20 04 82 47 65
%%%           19 01 23 75 03 34
%%%          88 02 77 73 07 63 67
%%%         99 65 04 28 06 16 70 92
%%%        41 41 26 56 83 40 80 70 33
%%%       41 48 72 33 47 32 37 16 94 29
%%%      53 71 44 65 25 43 91 52 97 51 14
%%%     70 11 33 28 77 73 17 78 39 68 17 57
%%%    91 71 52 38 17 14 91 43 58 50 27 29 48
%%%   63 66 04 68 89 53 67 30 73 16 69 87 40 31
%%%  04 62 98 27 23 09 70 98 73 93 38 53 60 04 23
%%%
%%% NOTE: As there are only 16384 routes, it is possible to solve this 
%%% problem by trying every route. However, Problem 67, is the same 
%%% challenge with a triangle containing one-hundred rows; it cannot be 
%%% solved by brute force, and requires a clever method! ;o)
%%%
%%% Created :  5 Dec 2008
%%%-------------------------------------------------------------------
-module(p18).

%% API
-compile(export_all).

%%====================================================================
%% API
%%====================================================================

%Build the tree from the bottom up,
%so that each node contains its the two children nodes
%and the sum of the node with the larger child.
%When we reach the root its sum will be the maximum total that we are 
%looking for.

%%--------------------------------------------------------------------
%% Function: solution(TreeRows) -> int()
%% Description:Returns the largest sum from the top to the bottom.
%%--------------------------------------------------------------------
solution(TreeRows) ->
    {_Root, _TreeL, _TreeR, RootW} = make_tree(lists:reverse(TreeRows), []),
    RootW.

test()->
    Rows = 
        [[75],
         [95, 64],
         [17, 47, 82],
         [18, 35, 87, 10],
         [20,  4, 82, 47, 65],
         [19,  1, 23, 75,  3, 34],
         [88,  2, 77, 73,  7, 63, 67],
         [99, 65,  4, 28,  6, 16, 70, 92],
         [41, 41, 26, 56, 83, 40, 80, 70, 33],
         [41, 48, 72, 33, 47, 32, 37, 16, 94, 29],
         [53, 71, 44, 65, 25, 43, 91, 52, 97, 51, 14],
         [70, 11, 33, 28, 77, 73, 17, 78, 39, 68, 17, 57],
         [91, 71, 52, 38, 17, 14, 91, 43, 58, 50, 27, 29, 48],
         [63, 66,  4, 68, 89, 53, 67, 30, 73, 16, 69, 87, 40, 31],
         [ 4, 62, 98, 27, 23,  9, 70, 98, 73, 93, 38, 53, 60,  4, 23]],
    solution(Rows).
    
test1() ->
    Rows = [[3], 
     [7, 5],
     [2, 4, 6],
     [8, 5, 9, 3]],
    solution(Rows).

%%====================================================================
%% Internal functions
%%==================================================================== 
%Builds a tree from a list of rows (in reversed order) 
make_tree([], [Root]) -> Root;
make_tree([H|T], []) ->
    make_tree(T, [make_leaf(V) || V <- H]);
make_tree([H|T], Acc) ->
    make_tree(T, make_row(H, Acc, [])).

%builds a row from a list
make_row([], _, Acc) ->
    lists:reverse(Acc);
make_row([RootH|RootT], [SubH1|[SubH2|SubT]], Acc) ->
    make_row(RootT, [SubH2|SubT], [make_node(RootH, SubH1, SubH2)| Acc]). 

%builds a node in the formst {Root, Child1, Child2, Root + max(Child1, Child2)}
make_node(Root, Child1, Child2)->
    Left = element(4, Child1),
    Right = element(4, Child2),
    {Root, Child1, Child2, Root +  case Left > Right of
                                    true -> Left;
                                    _ -> Right
                                 end}.

%builds a tree leaf
make_leaf(Value) ->
    {Value, {}, {}, Value}.
    

